FINAL REPORT

NASA GRANT NAG3-368

1 October 1982 - 30 September 1986

The Temperature Dependence of Inelastic Light Scattering from Small Particles for Use in Combustion Diagnostic Instrumentation

prepared by

Stanley D. Cloud
Associate Professor of Physics
University of Nevada, Las Vegas
Las Vegas, Nevada 89154

submitted to

The National Aeronautics & Space Administration
Lewis Research Center
Cleveland, Ohio

October 9, 1987

I. INTRODUCTION

The work described in this report falls into two parts: (1) a computer calculation of the expected angular distribution of coherent anti-Stokes Raman scattering (CARS) from micrometer size polystyrene spheres based on a Mie-type model, and (2) a pilot experiment to test the feasibility of measuring CARS angular distributions from micrometer size polystyrene spheres by simply suspending them in water. The computer calculations predict very interesting structure in the angular distributions that depends strongly on the size and relative refractive index of the spheres. The expected sharp structure makes an experiment particularly important not only as a test of the theoretical model's details but also as preliminary to a variety of possible applications.

This work grew out of a project to find means for increasing the sensitivity or applicability of CARS as a tool for combustion diagnostic instrumentation. A great deal of work has been done in the application of CARS to combustion studies[1,2], yet many problems remain to be solved. The theoretical work of Cooney and Gross[3] in 1982 showed that CARS from small particles might be enhanced by an order of magnitude over the intensity of scattering from the bulk material and by an even greater amount over the intensity of scattering from gas molecules. This suggested the possibility of seeding a flame with particles of suitable material and using CARS from the particles as a probe of local gas density and local pressure.

Since only Cooney's and Gross's computation of CARS from benzene droplets had been made and since no experiments had been performed at all, it was essential to develop a basis of scientific information about particle CARS before passing judgment on the technique as a combustion diagnostics instrumentation tool. The first step was to extend the computational capability to include any material and to produce complete angular information about CARS from a particle of any size within the limits of the Mie-type calculation. The second step was to select a convenient particle material and attempt to measure the CARS angular distribution for comparison with the calculational model.

II. THEORETICAL CALCULATIONS

The first reported computer calculations of CARS from micrometer size particles were by Cooney and Gross[3]. Due to limitations of computer facilities, they only considered CARS from benzene liquid droplets, and they only did computations for thirteen different droplet sizes. Earlier experiments[4] with fluorescence from dye-doped latex spheres had shown sharp "size resonances" that might appear in CARS also. In order to indicate

the presence of such resonances, which are less than one percent in width, calculations needed to be made for a range of closely spaced particle sizes. The Cray-1S supercomputer at the Lewis Research Center made such detailed calculations possible.

Details of the Mie-type model used for the computation of CARS angular distributions from particles are contained in the Ph.D. dissertation of Gross.[5] The essential ideas are the following. In CARS, three incident photons interact simultaneously with a molecule to produce a fourth photon with a different energy. Typically, two of the incident photons are frequency-doubled Nd:YAG laser photons with a wavelength of 532nm, while the third incident photon comes from a tunable dye laser driven also by the Nd:YAG laser. The dye laser wavelength is chosen in accordance with the Raman resonance one desires to use in a particular material. If the energy of the Nd:YAG laser photons is E_1 and the energy of the dye laser photon is E_2, then the fourth photon resulting from CARS will have energy $2E_1-E_2$.

From the point of view of electric fields, in the CARS process described above, two of the incident photons induce fields in the particle that oscillate at frequency f_1 while the third induces a field oscillating at frequency f_2. The presence of these fields simultaneously superimposed in the material produces a new electromagnetic oscillation at a frequency f_3 equal to $2f_1-f_2$. The new frequency results because the material has a third-order, nonlinear susceptibility.

In order to predict the intensity and directional distribution of the new CARS produced light theoretically, the oscillation at frequency $f_3=2f_2-f_1$ is introduced as a source into Maxwell's equations. Maxwell's equations combine to form a wave equation with a source term at frequency f_3 whose intensity is a function of position within the particle. The wave equation can be solved numerically by the Green function technique and multipole expansion, subject to boundary conditions at the surface of the sphere. A mathematical outline of the procedure is given in appendix A.

The computation routine involves several steps. First, the field distributions at frequencies f_1 and f_2 are found inside the sphere in terms of the amplitudes and polarizations of the incident photon beams. To do this, the fields are expressed in terms of Ricatti-Bessel functions and associated Legendre functions. Then the distribution of sources at f_3 is calculated and integrated with the vector multipoles over the volume of the particle. The vector multipoles are computed from spherical Bessel functions and irreducible tensors, which are expressed in terms of spherical harmonics by means of the Clebsch-Gordan coefficients. Next, the integrals over the volume are combined with the boundary conditions at the surface of the sphere to yield coefficients in a multipole expansion of the scattered field. The scattered field is computed angle by angle as the partial sum of the expansion, including enough terms to give sufficient precision in the final result.

The program is exhibited in appendix B. It is very simple to use, because very little input data are required. The size parameter of the particle is input at line 3300, and the number of terms to be included in the multipole expansion at line 3400. The three wavelengths and the corresponding refractive indices are entered directly into the common block as REFRAC(3) and WAVE1(3). The entered wavelengths are those outside the particle. The program computes the wavenumbers inside the particle at line 4000.

In operation, the program begins calculating the incident fields inside the particle using subroutines MIECO and INFLD. MIECO is called at line 4200. The special functions are called, and then at line 6400, the integration of the new field over the particle volume is begun. DPROD (line 8700) is the new field at f_3. The resulting integrals are CLE and CLM, and they are the coefficients to be used in the multipole sums begun in line 14200. Output of the program is an angular distribution of the logarithm of the CARS intensity in relative units, printed at line 18600. FIT is the scattering angle in degrees measured from the direction of the incident beams. DCS is the log intensity.

The subroutines used are:

MIECO	calculates the Mie-type coefficients for use in finding the internal fields in INFLD
RBES	computes the Ricatti-Bessel functions;
RHANK	computes the Ricatti-Hankel functions;
INFLD	finds the internal fields;
LEG	calculates the Legendre functions;
YLM	computes the spherical harmonics;
TLM	computes the irreducible tensors or vector spherical harmonics;
CLG	calculates the Clebsch-Gordan coefficients.

Note that wherever possible the program is vectorized to run with optimal speed on a Cray-1 computer. For LMAX = 20, which is sufficient for size parameters up to about 8, the program runs in approximately 40 sec of CPU time. Run time increases rapidly with increasing LMAX. With LMAX = 95, the program took about 3.3 hours.

CARS angular distributions were computed on the NASA Lewis Research Center Cray-1S computer, transmitted via Telenet to an IBM-PC, and stored on floppy disks. The angular distributions were then sent by telephone line from the IBM-PC to the University of Nevada, Las Vegas Computing Center Cyber-172

computer where they were archived and plotted using the National Center for Atmospheric Research graphics package called NCAR.

The first calculations were a set of angular distributions for 29 different size parameters ranging from 5.39 to 6.79 for a polystyrene sphere immersed in air. These results are presented in figures 1, 2, and 3. The vertical scale is the logarithm of the scattered intensity on a relative scale.

Polystyrene was chosen as the material in anticipation of an experiment. Polystyrene is a polymer of styrene, whose structure includes a benzene ring. So polystyrene has Raman scattering properties very similar to benzene; in particular, it exhibits a strong, isolated Raman resonance near 1000 cm^{-1}.[6] Polystyrene spheres carefully manufactured in a range of sizes are available commercially as "uniform latex".

Figures 1 through 3 show angular distributions with considerable structure that changes rapidly at certain scattering angles as the particle size changes. The change of structure with size is a property that may be exploitable for applications. This will be discussed further in the section entitled "Discussion". However, the figures show that the change of angular distribution with size is very complicated, so great detail will be required for applications.

Figure 4 shows the variation of CARS angular distributions as the refractive index changes. The refractive index is, of course, the relative refractive index, so a change can be the result of something happening within the particle or in the surrounding medium. The figure shows that CARS scattering changes relatively slowly with refractive index. Figure 5 shows that there is practically no change in CARS angular distribution when the ambient air pressure is increased from zero to 90 atmospheres.

For the purpose of doing an experiment to compare actual CARS with the computed angular distributions, one wants a scatterer that will produce structured distributions that are not sensitive to experimental conditions. For example, in a sample of polystyrene latex there might be a variation of 1% or 2% in the sizes of the particles. If the scattering is sensitive to such small changes, then the particle distribution will smear the angular distributions. Providing the size distribution is known, it is possible to integrate over the range of angular distributions to obtain a smeared theoretical distribution. However, one should expect violently changing conditions in the vicinity of the high-powered laser beam required for CARS, and both the nature and the magnitude of the changes may be difficult to determine precisely. For example, due to heating or ablation, the actual size of the particles might change appreciably. Furthermore, the surrounding medium could expand or even vaporize if it is a liquid. In general, when the relative refractive index is reduced, the amount of structure and the size-variation of the structure in CARS angular distributions is

reduced. This is seen in figures 6 and 7, which present CARS in water for a range of size parameters and for a range of refractive indices. It suggests that for a preliminary experiment, it would be desirable to suspend the particles in a medium, such as water, whose refractive index is different from that of polystyrene (approximately 1.6), so that some structure will be present. But the surrounding refractive index should not be too different. A relative refractive index of about 1.2, as shown in the figures, should give good results.

These computations were presented in a conference paper.[7]

III. EXPERIMENT

Preliminary experiments to observe CARS from polystyrene latex spheres were carried out using the Nd:YAG/Dye laser CARS system at the NASA Lewis Research Center. The plan included generating CARS from latex suspended in water held in a cell within a Wyatt Technology "Dawn Model B" fifteen-detector light scattering system.[8] The fifteen detectors were intended to detect the scattered light at fifteen angles simultaneously. Output from the detectors would be digitized and stored in a personal computer for later analysis. Interference filters[9] to transmit the CARS generated light but block directly scattered incident light were fitted in front of the fifteen detectors. Although the experiment did not proceed to the point of producing CARS within the "Dawn" system, the "Dawn" system was interfaced to the computer and tested satisfactorily as a detector system. Furthermore, the interference filters were found to discriminate very well against light from the laser beams, transmitted the CARS wavelengths as expected, and were extensively used in the pilot CARS study to be described.

Before attempting to observe CARS from polystyrene particles, an effort was made to see CARS from bulk liquid benzene. A quartz cylindrical cell filled with benzene was positioned with its flat end faces perpendicular to the approximately colinear incident beams. With the dye laser wavelength set at about 562nm and its intensity roughly one-half the intensity of the Nd:YAG laser beam at 532nm, a spot of blue light could be seen slightly off axis when the light exiting the cell was allowed to fall on a white card. It was necessary to view the card through the interference filters mentioned above in order to suppress the incident laser beams and also the yellow and red Raman shifted laser beams. The blue light was directed into a spectrometer via a prism, and the wavelength was measured to be about 505nm in agreement with the $f_3=2f_2-f_1$ expectation.

To explore the possibility of detecting CARS from polystyrene microspheres, a very dilute suspension of 1.091 micrometer latex particles[10] in water was prepared. The concentration of polystyrene in the cell was about 5×10^{-7}, which would give approximately 5 microspheres in the interaction region where the incident laser beams were focused. The dimensions of the

interaction region were estimated to be 1mm of length by 0.1mm diameter. Using the interference filters to block scattered incident laser light, it was possible to observe visually blue light emerging from the region of beam focus. A fiber bundle was used to collect the blue light and direct it through a lens that focused it on the slit of a double spectrometer. To confirm that it was CARS from the polystyrene spheres, the light passing through the spectrometer was analyzed by an optical multichannel analyzer system gated synchronously with laser pulses. As expected, a wavelength peak was found at about 505nm. However, the 505nm light remained present as the dye laser wavelength was scanned over a 10nm or more range, whereas a peak with a width of perhaps $20cm^{-1}$ (see ref. 6) or about 0.6nm was expected. This "continuous" background was attributed to weak CARS from a distant Raman resonance in water. To detect the presence of polystyrene, then, it was necessary to search for a weak, narrow peak superimposed on the water background in a plot of 505nm light intensity versus dye laser wavelength.

Data collection was quite difficult because it was necessary to keep the laser beam intensities down so as to minimize damage to the cell window and also prevent boiling of the water. This meant that the signal was weak and difficult to collect efficiently with the optical fiber bundle. Integration periods of several minutes were required, during which the laser output power could drift, so laser power was monitored and regulated manually. Furthermore, whenever window damage was detected by the presence of photoacoustic pulses, the cell had to be moved slightly and the fiber bundle had to be repositioned. Average laser power was typically 13mW at 532nm and 6mw at 562nm with a 10 hertz repetition rate.

Late on the next to last day of scheduled time, just such a narrow peak was found in the day's last run. The data are shown in figure 8. Both peak position and width are as expected. Unfortunately, due to the difficulties mentioned above, it was not possible to confirm the presence of the peak.

DISCUSSION

The theoretical results indicate that CARS from particles is rich in structure. Whether the structure can be exploited efficiently for instrumentation purposes remains to be determined, and much more developmental work is needed. In particular, the fundamental question of the applicability of the Mie-type model to droplets of benzene or to polystyrene spheres has not yet been answered experimentally. The particle's surface is included in the model as an interface at which boundary conditions must match. The matching of boundary conditions requires the addition of a field in addition to that directly attributable to the sources at f_3, and that field results from the surface. But the model does not take into account that the CARS interaction might be different at the surface or that surface currents might be produced. Especially with such small

particles with large surface to volume ratios under the influence of very intense incident fields, surface effects may be significant. The Yale group has been successful in observing nonlinear scattering from droplets recently[11], but they are working with larger droplets, of about 30 micrometers, rather than the approximately 1 micrometer particles considered here.

Some modifications in the apparatus should be made before further experiments are attempted. If possible, laser power stability should be improved. In any case, the scattered light collection system should be made more efficient. Perhaps CARS signal strength can be increased by using a dense aerosol of particles instead of a dilute suspension.

Although the experiment described here was conceived more optimistically than it turned out, it was nevertheless intended only as a pilot study to determine the feasibility of detecting CARS from small particles and to discover what the main barriers to success are likely to be. It was successful in those respects: CARS was observed, if only briefly from polystyrene particles; and a clear idea of the difficulties has emerged.

ACKNOWLEDGMENTS

I owe special thanks to Dr. Nancy Piltch, who very generously interrupted her own research to make her CARS facility available to me for a month, who patiently instructed me in its use, and who assisted with the experiments. Mr. Ken Weiland provided excellent and absolutely essential technical support. Dr. Arthur Decker first interested me in this project and, with Mr. Daniel Lesco, Mr. William Nieberding, and Mr. Norman Wenger, provided encouragement and support.

V. REFERENCES

1. Eesley, G. L., Coherent Raman Spectroscopy (Pergamon Press, 1981).

2. Hall, Robert J., and Alan C. Eckbreth, "Coherent Anti-Stokes Raman Spectroscopy (CARS): Application to Combustion Diagnostics," in Laser Applications Volume 5, ed. by John F. Ready and Robert K. Erf (Academic Press, Inc. 1984).

3. Cooney, John and Abraham Gross, "Coherent anti-Stokes Raman scattering by droplets in the Mie size range," Optics Letters 7, 218 (1982).

4. Benner, R. E., P. W. Barber, J. F. Owen, and R. K. Chang, "Observations of Structure Resonances in the Fluorescence Spectra from Microspheres," Physical Review Letters 44,

475 (1980).

5. Gross, Abraham, Coherent Anti-Stokes Raman Scattering by Droplets of the Mie Size Range, Drexel University Ph.D. Dissertation, 1982.

6. Hetherington, W. M. III, N. E. Van Wyck, E. W. Koenig, G. I. Stegeman, and R. M. Fortenberry, "Observation of coherent Raman scattering in thin-film optical waveguides," Optics Letters 9, 88 (1984).

7. Cloud, Stanley D., and David Wruck, "Size and Refractive Index Dependence of Coherent Anti-Stokes Raman Scattering from Micrometer-Size Polystyrene Spheres," Optics News 10, No. 5, 72 (1984).

8. Wyatt Technology Corporation, 820 East Haley Street, P.O. Box 3003, Santa Barbara California, 93130.

9. Omega Optical, Inc., 3 Grove Street, P.O. Box 573, Brattleboro, Vermont 05301

10. Duke Scientific Corporation, 2415 Embarcadero Way, Palo Alto, California 94303.

11. For example, Snow, Judith B., Shi-Xiong Qian, and Richard K. Chang, "Stimulated Raman scattering from individual water and ethanol droplets at morphology-dependent resonances," Optics Letters 10, 37 (1985).

APPENDIX A

Mathematical Outline of the Procedure for Computing CARS from Spherical Dielectric Particles

I. The outgoing CARS power per unit solid angle is computed from the electromagnetic field amplitude, H:

$$\frac{dP}{d\Omega}(\Theta,\Phi) = \frac{c}{8\pi} r^2 [\vec{H}_3 \cdot \vec{H}_3^*], \text{ where}$$

II. H is computed as an expansion of irreducible tensors:

$$\vec{H}_3 = \frac{e^{ik3r}}{r} \sum_l (-i)^{l+1} \sum_m CM(l,m)(\vec{r} \times \vec{T}_{llm}) + CE(l,m)\vec{T}_{llm}$$

$$\vec{T}_{llm} = \sum_\mu C(lll;m-\mu,\mu) Y_l^{m-\mu}(\Theta,\Phi) \hat{\xi}_\mu \quad ,$$

III. and the coefficients are:

$$CM(l,m) = - \frac{4\pi n_3 k_3^3 \int \vec{P}(3) \cdot \vec{A}_{lm}^{in(M)} \, d^3r'}{(xh)'(yj) - n_3(yj)'(xh)}$$

$$CE(l,m) = \frac{4\pi n_3 k_3^3 \int P(3) \cdot \vec{A}_{lm}^{in(E)} \, d^3r'}{n_3 (xh)'(yj) - (yj)'(xh)} \quad ,$$

with $$\vec{A}_{lm}^{in(M)} = j_l(n_3 k_3 r) \vec{T}_{llm}(\Theta,\Phi)$$

$$A_{lm}^{in(E)} = - \frac{i}{n_3 k_3} \vec{\nabla} \times \vec{A}_{lm}^{in(M)},$$

and $$\vec{P}(3) = \chi \vec{E}(f_1)\vec{E}(f_2)\vec{E}(f_3).$$

n is the refractive index, k the wave number. h and j are the spherical Hankel and Bessel functions. x is $k_3 a$, y is $n_3 x$, and a is the particle radius.

APPENDIX B

THE CRAY-1 PROGRAM "SCARS"

```
0000800       PROGRAM SCARS
0000900       COMMON/PROPS/PI,PMB,RA,REFRAC(3),WAVE1(3),WAVE2(3),LMAX
0001000       COMMON/COORDS/R,COST,SINT,COSP,SINP,P
0001100       COMMON/FIELDS/ER(2),ET(2),EP(2)
0001200       COMMON/RAD2/RB(100),DRB(100)
0001300       COMMON/RAD3/RH(100),DRH(100)
0001400       COMMON/RAD4/RB2(100),DRB2(100)
0001500       COMMON/THETA/PIE(100),TAU(100)
0001600       COMMON/COEFS1/AMIE(100,2),BMIE(100,2)
0001700       COMMON/COEFS2/CMIE(100,2),DMIE(100,2)
0001800       COMMON/HAR/Y(100,100)
0001900       COMMON/CGC/C(3,3,100,100)
0002000       COMMON/TEN/TB(3,3,100,100)
0002100       COMPLEX ER,ET,EP
0002200       COMPLEX TB
0002300       COMPLEX RH,DRH
0002400       COMPLEX AMIE,BMIE
0002500       COMPLEX CMIE,DMIE
0002600       COMPLEX Y
0002700       COMPLEX SM1(3),SM2(3),SE(3),SDB(3)
0002800       COMPLEX DM(100),DE(100)
0002900       COMPLEX CLM(100,100),CLE(100,100)
0003000       COMPLEX AM(3,100,100),AE(3,100,100),APM(100,100),APE(100,100)
0003100       COMPLEX DPROD,IMAGI,CIL
0003200       DIMENSION VL(100),VPL(100)
0003300       SIZEP=70.0
0003400       LMAX=95
0003500       LLMAX=LMAX-1
0003600       CALL CLG
0003700       DO 1 I=1,3
0003800       WAVE1(I)=2.0*PI/WAVE1(I)
0003900 C        ABOVE CONVERTS INPUT WAVELENGTHS TO WAVENUMBERS
0004000     1 WAVE2(I)=WAVE1(I)*REFRAC(I)
0004100       RA=SIZEP/WAVE1(1)
0004200       CALL MIECO(1)
0004300       CALL MIECO(2)
0004400 C   SAVE CARS 'INSIDE' RICATTI BES. FUNCS. AS RB2 & DRB2
0004500       ARG=WAVE2(3)*RA
0004600       CALL RBES(ARG)
0004700       DO 2 L=1,LMAX
0004800       RB2(L)=RB(L)
0004900       DRB2(L)=DRB(L)
0005000       DL=1.0/FLOAT(L+L+1)
0005100       FL=FLOAT(L)
0005200       VL(L)=SQRT(FL*DL)
0005300       VPL(L)=(FL+1.)*DL
0005400       VPL(L)=SQRT(VPL(L))
0005500     2 CONTINUE
0005600 C   CREATE CARS 'OUTSIDE' RIC. HANK. FUNCS.
0005700       ARG=WAVE1(3)*RA
0005800       CALL RBES(ARG)
0005900       CALL RHANK(ARG)
0006000  3000 FORMAT(18H BEGIN INTEGRATION)
0006100 C
0006200 CBEGIN INTEGRATION TO CREATE CLM(L,M) & CLE(L,M)
0006300 C
```

```
0006400       DO 10 I1=21,40
0006500  3100 FORMAT(5H I1= ,I5)
0006600       X=0.05*FLOAT(I1)-1.025
0006700       XS=X*X
0006800       DO 10 I2=21,40
0006900  3200 FORMAT(5H I2= ,I5)
0007000       CY=0.05*FLOAT(I2)-1.025
0007100       YS=CY*CY
0007200       DO 10 I3=1,40
0007300       Z=0.05*FLOAT(I3)-1.025
0007400       ZS=Z*Z
0007500       RS=XS+YS+ZS
0007600       IF(RS.GT.0.998) GOTO 10
0007700       IF(RS.LT.1.E-4) GOTO 10
0007800       R=SQRT(RS)
0007900       COST=Z/R
0008000       RO=SQRT(XS+YS)
0008100       SINT=RO/R
0008200       COSP=X/RO
0008300       SINP=CY/RO
0008400       P=ASIN(SINP)
0008500       CALL INFLD(1)
0008600       CALL INFLD(2)
0008700       DPROD=ER(1)*CONJG(ER(2))+ET(1)*CONJG(ET(2))+EP(1)*CONJG(EP(2))
0008800 C     GENERATE A(L,M)S
0008900       ARG=WAVE2(3)*R*RA
0009000       CALL RBES(ARG)
0009100       DO 3 LD=1,LMAX
0009200     3 RB(LD)=RB(LD)/ARG
0009300 C        ABOVE MADE SPHER. BESS. FUNC. FROM RICATTI B. F.
0009400       RB0=SIN(ARG)/ARG
0009500       CALL YLM
0009600       CALL TLM
0009700 C  AE & AM HERE ARE MULTIPOLE VECTORS, NOT COEFFS.
0009800       DO 5 IC=1,3
0009900       AM(IC,1,1)=TB(IC,2,1,1)*RB(1)
0010000       AM(IC,1,2)=TB(IC,2,1,2)*RB(1)
0010100       AE(IC,1,1)=-VL(1)*TB(IC,3,1,1)*RB(2)+VPL(1)*TB(IC,1,1,1)*RB0
0010200     5 AE(IC,1,2)=-VL(1)*TB(IC,3,1,2)*RB(2)+VPL(1)*TB(IC,1,1,2)*RB0
0010300       DO 9 L=2,LLMAX
0010400       MTOP=L+1
0010500       DO 8 M=1,MTOP
0010600       AM(1,L,M)=TB(1,2,L,M)*RB(L)
0010700       AM(2,L,M)=TB(2,2,L,M)*RB(L)
0010800       AM(3,L,M)=TB(3,2,L,M)*RB(L)
0010900       AE(1,L,M)=-VL(L)*TB(1,3,L,M)*RB(L+1)+VPL(L)*TB(1,1,L,M)*RB(L-1)
0011000       AE(2,L,M)=-VL(L)*TB(2,3,L,M)*RB(L+1)+VPL(L)*TB(2,1,L,M)*RB(L-1)
0011100       AE(3,L,M)=-VL(L)*TB(3,3,L,M)*RB(L+1)+VPL(L)*TB(3,1,L,M)*RB(L-1)
0011200     8 CONTINUE
0011300     9 CONTINUE
0011400 C BEGIN NEW L LOOP STARTING WITH L=1
0011500       DO 19 L=1,LLMAX
0011600       MTOP=L+1
0011700       DO 18 M=1,MTOP
0011800       AM(1,L,M)=CONJG(AM(1,L,M))
0011900       AM(2,L,M)=CONJG(AM(2,L,M))
0012000       AM(3,L,M)=CONJG(AM(3,L,M))
0012100       AE(1,L,M)=CONJG(AE(1,L,M))
0012200       AE(2,L,M)=CONJG(AE(2,L,M))
0012300       AE(3,L,M)=CONJG(AE(3,L,M))
0012400    18 CONTINUE
0012500       DO 28 M=1,MTOP
0012600       APM(L,M)=(AM(1,L,M)*ER(1)+AM(2,L,M)*ET(1)+AM(3,L,M)*EP(1))*DPROD
0012700       APE(L,M)=(AE(1,L,M)*ER(1)+AE(2,L,M)*ET(1)+AE(3,L,M)*EP(1))*DPROD
0012800    28 CONTINUE
0012900 C   CLM & CLE ARE THEINTEGRALS CALLED SIGMA(M) & SIGMA(E) IN NOTES
```

```
0013000        DO 38 M=1,MTOP
0013100        CLM(L,M)=CLM(L,M)+APM(L,M)
0013200        CLE(L,M)=CLE(L,M)+APE(L,M)
0013300     38 CONTINUE
0013400     19 CONTINUE
0013500     10 CONTINUE
0013600 C       CLE(L,M)=CONJG(CLE(L,M))
0013700 C      CLM(L,M)=CONJG(CLM(L,M))
0013800   1500 FORMAT(E15.7,3X,E15.7)
0013900 C
0014000 C     BEGIN S SUM FOR EACH OBSERVATION ANGLE
0014100 C
0014200        IMAGI=(0.0,1.0)
0014300        SINP=1.0
0014400        COSP=0.0
0014500        P=PI/2.
0014600        WRITE(12,2000)
0014700        DO 30 L=1,LLMAX
0014800        DM(L)=DRH(L)*RB2(L)-REFRAC(3)*DRB2(L)*RH(L)
0014900     30 DE(L)=REFRAC(3)*DRH(L)*RB2(L)-RH(L)*DRB2(L)
0015000        DO 70 IT=1,179
0015100        FIT =FLOAT(IT)
0015200        T=FIT*PI/180.
0015300        SINT=SIN(T)
0015400        COST=COS(T)
0015500        DO 40 IC=1,3
0015600        SDB(IC)=(0.0,0.0)
0015700     40 CONTINUE
0015800        CIL=-IMAGI
0015900        CALL YLM
0016000        CALL TLM
0016100        DO 60 L=1,LLMAX
0016200        DO 55 IC=1,3
0016300        SM1(IC)=(0.0,0.0)
0016400        SM2(IC)=(0.0,0.0)
0016500     55 SE(IC)=(0.0,0.0)
0016600        MTOP=L+1
0016700        DO 50 M=2,MTOP,2
0016800        SM1(1)=SM1(1)+CLM(L,M)*AIMAG(TB(1,3,L,M))
0016900        SM1(2)=SM1(2)+CLM(L,M)*AIMAG(TB(2,3,L,M))
0017000        SM1(3)=SM1(3)+CLM(L,M)*AIMAG(TB(3,3,L,M))
0017100        SM2(1)=SM2(1)+CLM(L,M)*AIMAG(TB(1,1,L,M))
0017200        SM2(2)=SM2(2)+CLM(L,M)*AIMAG(TB(2,1,L,M))
0017300        SM2(3)=SM2(3)+CLM(L,M)*AIMAG(TB(3,1,L,M))
0017400        SE(1)=SE(1)+CLE(L,M)*AIMAG(TB(1,2,L,M))
0017500        SE(2)=SE(2)+CLE(L,M)*AIMAG(TB(2,2,L,M))
0017600        SE(3)=SE(3)+CLE(L,M)*AIMAG(TB(3,2,L,M))
0017700     50 CONTINUE
0017800        CIL=-CIL*IMAGI
0017900        DO 65 IC=1,3
0018000     65 SDB(IC)=SDB(IC)+CIL*((VL(L)*SM1(IC)+VPL(L)*SM2(IC))/DM(L)-IMAGI*
0018100       CSE(IC)/DE(L))
0018200     60 CONTINUE
0018300   1510 FORMAT(6(E15.7,3X))
0018400        DCS=SDB(1)*CONJG(SDB(1))+SDB(2)*CONJG(SDB(2))+SDB(3)*CONJG(SDB(3))
0018500        DCS=ALOG10(DCS)
0018600        WRITE(12,2100)FIT,DCS
0018700     70 CONTINUE
0018800   1000 FORMAT(F7.3,1X,I3)
0018900   1100 FORMAT(32H TYPE SIZEP(F7.3) SPACE LMAX(I3))
0019000   2000 FORMAT(16H(E15.7,3X,E15.7))
0019100   2100 FORMAT(E15.7,3X,E15.7)
0019200        STOP
0019300        END
0019400        SUBROUTINE MIECO(IN)
0019500        COMMON/PROPS/PI,PMR,RA,REFRAC(3),WAVE1(3),WAVE2(3),LMAX
```

```
0019600       COMMON/RAD2/RB(100),DRB(100)
0019700       COMMON/RAD3/RH(100),DRH(100)
0019800       COMMON/RAD4/RB2(100),DRB2(100)
0019900       COMMON/COEFS1/AMIE(100,2),BMIE(100,2)
0020000       COMMON/COEFS2/CMIE(100,2),DMIE(100,2)
0020100       COMPLEX RH,DRH,AMIE,BMIE,CMIE,DMIE
0020200       X=WAVE2(IN)*RA
0020300       CALL RBES(X)
0020400       DO 10 L=1,LMAX
0020500         RB2(L)=RB(L)
0020600         DRB2(L)=DRB(L)
0020700    10 CONTINUE
0020800       X=WAVE1(IN)*RA
0020900       CALL RBES(X)
0021000       CALL RHANK(X)
0021100       XXX=REFRAC(IN)*PMB
0021200       DO 20 L=1,LMAX
0021300         AMIE(L,IN)=1.0/(PMB*RH(L)*DRB2(L)-REFRAC(IN)*DRH(L)*RB2(L))
0021400         DMIE(L,IN)=1.0/(PMB*DRH(L)*RB2(L)-REFRAC(IN)*RH(L)*DRB2(L))
0021500         BMIE(L,IN)=(REFRAC(IN)*RB(L)*DRB2(L)-PMB*DRB(L)*RB2(L))*      -
0021600      CDMIE(L,IN)
0021700         CMIE(L,IN)=XXX*(RB(L)*DRH(L)-RH(L)*DRB(L))
0021800         DMIE(L,IN)=CMIE(L,IN)*DMIE(L,IN)
0021900         CMIE(L,IN)=-AMIE(L,IN)*CMIE(L,IN)
0022000         AMIE(L,IN)=(REFRAC(IN)*DRB(L)*RB2(L)-PMB*RB(L)*DRB2(L))*      -
0022100      CAMIE(L,IN)
0022200    20 CONTINUE
0022300       RETURN
0022400       END
0022500       SUBROUTINE RBES(X)
0022600       COMMON/PROPS/PI,PMB,RA,REFRAC(3),WAVE1(3),WAVE2(3),LMAX
0022700       COMMON/RAD2/RB(100),DRB(100)
0022800       S=SIN(X)
0022900       C=COS(X)
0023000       RIC1=S/X-C
0023100       IF(X.LT.0.9)GO TO 25
0023200       LBIG=LMAX+10
0023300       LBM1=LBIG-1
0023400       RB(LBIG)=0.1
0023500       RB(LBM1)=0.2
0023600       DO 10 L=2,LBM1
0023700         LL=LBIG-L
0023800         LLP1=LL+1
0023900         LLP2=LL+2
0024000         REALIE=2*LL+3
0024100         RB(LL)=REALIE*RB(LLP1)/X-RB(LLP2)
0024200    10 CONTINUE
0024300       SCALE=RIC1/RB(1)
0024400       RB(1)=RIC1
0024500       DRB(1)=S-RB(1)/X
0024600       DO 20 L=2,LMAX
0024700         RB(L)=SCALE*RB(L)
0024800         LM1=L-1
0024900         REALIE=L
0025000         DRB(L)=RB(LM1)-REALIE*RB(L)/X
0025100    20 CONTINUE
0025200       RETURN
0025300    25 RB(1)=RIC1
0025400       RB(2)=3.0*RB(1)/X-S
0025500       DRB(1)=S-RB(1)/X
0025600       DRB(2)=RB(1)-2.0*RB(2)/X
0025700       FRONT=X**3/15.0
0025800       Z=0.5*X*X
0025900       DO 40 L=3,LMAX
0026000         REALIE=2*L+1
0026100         FRONT=FRONT*X/REALIE
```

```
0026200         TERM=1.0
0026300         SERIES=1.0
0026400         NUM=2*L+3
0026500         DO 30 I=1,6
0026600           REALIE =NUM*I
0026700           TERM=-TERM*Z/REALIE
0026800           SERIES=SERIES+TERM
0026900           NUM=NUM+2
0027000     30    CONTINUE
0027100         RB(L)=FRONT*SERIES
0027200         LM1=L-1
0027300         REALIE=L
0027400         DRB(L)=RB(LM1)-REALIE*RB(L)/X
0027500     40 CONTINUE
0027600        RETURN
0027700        END
0027800        SUBROUTINE RHANK(X)
0027900        COMMON/RAD3/RH(100),DRH(100)
0028000        COMMON/RAD2/RB(100),DRB(100)
0028100        COMMON/PROPS/PI,PMB,RA,REFRAC(3),WAVE1(3),WAVE2(3),LMAX
0028200        DIMENSION RN(100),DRN(100)
0028300        COMPLEX RH,DRH
0028400        S=SIN(X)
0028500        C=COS(X)
0028600        RN(1)=-C/X-S
0028700        RN(2)=3.0*RN(1)/X+C
0028800        DRN(1)=-C-RN(1)/X
0028900        DRN(2)=RN(1)-2.0*RN(2)/X
0029000        RH(1)=CMPLX(RB(1),RN(1))
0029100        RH(2)=CMPLX(RB(2),RN(2))
0029200        DRH(1)=CMPLX(DRB(1),DRN(1))
0029300        DRH(2)=CMPLX(DRB(2),DRN(2))
0029400        DO 10 L=3,LMAX
0029500          LM1=L-1
0029600          LM2=L-2
0029700          REALIE=2*L-1
0029800          RN(L)=REALIE*RN(LM1)/X-RN(LM2)
0029900          REALIE =L
0030000          DRN(L)=RN(LM1)-REALIE*RN(L)/X
0030100          RH(L)=CMPLX(RB(L),RN(L))
0030200          DRH(L)=CMPLX(DRB(L),DRN(L))
0030300     10 CONTINUE
0030400        RETURN
0030500        END
0030600        SUBROUTINE INFLD(NW)
0030700        COMMON/PROPS/PI,PMB,RA,REFRAC(3),WAVE1(3),WAVE2(3),LMAX
0030800        COMMON/COORDS/R,COST,SINT,COSP,SINP,P
0030900        COMMON/FIELDS/ER(2),ET(2),EP(2)
0031000        COMMON/RAD2/RB(100),DRB(100)
0031100        COMMON/THETA/PIE(100),TAU(100)
0031200        COMMON/COEFS2/CMIE(100,2),DMIE(100,2)
0031300        COMPLEX ER,ET,EP,CMIE,DMIE,C,IMAGI
0031400        IMAGI=(0.0,1.0)
0031500        RSIZE=WAVE2(NW)*R*RA
0031600        CALL RBES(RSIZE)
0031700        CALL LEG
0031800        ER(NW)=(0.0,0.0)
0031900        ET(NW)=(0.0,0.0)
0032000        EP(NW)=(0.0,0.0)
0032100        C=(1.0,0.0)
0032200        DO 10 L=1,LMAX
0032300          A=2*L+1
0032400          B=L*(L+1)
0032500          ER(NW)=ER(NW)+C*A*CMIE(L,NW)*RB(L)*PIE(L)*SINT
0032600          A=A/B
0032700          C=C*IMAGI
```

```
0032800         ET(NW)=ET(NW)+C*A*(DMIE(L,NW)*RB(L)*PIE(L)-IMAGI*CMIE(L,NW)*       -
0032900       CDRB(L)*TAU(L))
0033000         EP(NW)=EP(NW)+C*A*(IMAGI*CMIE(L,NW)*DRB(L)*PIE(L)-DMIE(L,NW)*      -
0033100       CRB(L)*TAU(L))
0033200      10 CONTINUE
0033300         ER(NW)=ER(NW)*COSP/(RSIZE**2)
0033400         ET(NW)=ET(NW)*COSP/RSIZE
0033500         EP(NW)=EP(NW)*SINP/RSIZE
0033600         RETURN
0033700         END
0033800         SUBROUTINE LEG
0033900         COMMON/PROPS/PI,PMB,RA,REFRAC(3),WAVE1(3),WAVE2(3),LMAX
0034000         COMMON/COORDS/R,X,SINT,COSP,SINP,P
0034100         COMMON /THETA/PIE(100),TAU(100)
0034200         PIE(1)=1.0
0034300         PIE(2)=3.0*X
0034400         TAU(1)=X
0034500         TAU(2)=6.0*X*X-3.0
0034600         DO 10 L=3,LMAX
0034700             LM1=L-1
0034800             LM2=L-2
0034900             REAL1=L
0035000             REAL2=LM1
0035100             REAL3=2*L-1
0035200             PIE(L)=(REAL3*X*PIE(LM1)-REAL1*PIE(LM2))/REAL2
0035300             REAL2=L+1
0035400             TAU(L)=REAL1*X*PIE(L)-REAL2*PIE(LM1)
0035500      10 CONTINUE
0035600         RETURN
0035700         END
0035800         SUBROUTINE YLM
0035900         COMMON/HAR/Y(100,100)
0036000         COMMON/COORDS/R,X,S,COSP,SINP,P
0036100         COMMON/PROPS/PI,PMB,RA,REFRAC(3),WAVE1(3),WAVE2(3),LMAX
0036200          COMPLEX Y
0036300         COMPLEX EIP(100)
0036400         DIMENSION YY(100,100)
0036500         DIMENSION PTM(100),SMP(100),CMP(100)
0036600         NMAX=LMAX+1
0036700         MMAX=NMAX+1
0036800         DO 1 I=1,NMAX
0036900         DO 1 J=1,MMAX
0037000       1 YY(I,J)=(0.0,0.0)
0037100         YY(1,2)=1.
0037200         YY(2,2)=X
0037300         YY(2,3)=-S/SQRT(2.)
0037400         YY(3,2)=(3.*X*X-1.)/2.
0037500         YY(3,3)=-SQRT(1.5)*X*S
0037600         YY(3,4)=SQRT(0.375)*S*S
0037700         DO 100 L=3,LMAX
0037800         FL=FLOAT(L)
0037900         RL=FL-1.
0038000         RLL=FL+RL
0038100         N=L+1
0038200         NL=N-1
0038300         IL=NL-1
0038400         YY(N,2)=(RLL*X*YY(NL,2)-RL*YY(IL,2))/FL
0038500         L2=L-2
0038600         DO 200 M=1,L2
0038700         NM=M+2
0038800         NMM=NM-1
0038900         FM=FLOAT(M)
0039000         PM=FL+FM
0039100         PMM=PM-1.
0039200         PM=FL-FM
```

```
0039400       YY(N,NM)=(SQRT(RM*RMM)*YY(IL,NM)-RLL*S*YY(NL,NMM))/SQRT(PM*PMM)
0039500   200 CONTINUE
0039600       L1=L-1
0039700       DO 300 M=L1,L
0039800       NM=M+2
0039900       NMM=NM-1
0040000       PM=FL+FLOAT(M)
0040100       PMM=PM-1.
0040200       YY(N,NM)=-RLL*S*YY(NL,NMM)/SQRT(PM*PMM)
0040300   300 CONTINUE
0040400   100 CONTINUE
0040500       FPI=0.5/SQRT(PI)
0040600       Y(1,2)=YY(1,2)*FPI
0040700       DO 500 L=1,LMAX
0040800       NP=L+2
0040900       RAD=FLOAT(L+L+1)
0041000       FACT=FPI*SQRT(RAD)
0041100       DO 800 NM=2,NP
0041200       PTM(NM)=FLOAT(NM-2)*P
0041300   800 CONTINUE
0041400       DO 700 NM=2,NP
0041500       SMP(NM)=SIN(PTM(NM))
0041600       CMP(NM)=COS(PTM(NM))
0041700   700 CONTINUE
0041800       DO 900 NM=2,NP
0041900       EIP(NM)=CMPLX(CMP(NM),SMP(NM))
0042000   900 CONTINUE
0042100       DO 400 NM=2,NP
0042200       Y(L+1,NM)=YY(L+1,NM)*EIP(NM)*FACT
0042300   400 CONTINUE
0042400   500 CONTINUE
0042500       DO 600 L=1,LMAX
0042600   600 Y(L+1,1)=-CONJG(Y(L+1,3))
0042700       RETURN
0042800       END
0042900       SUBROUTINE TLM
0043000       COMMON/PROPS/PI,PMB,RA,REFRAC(3),WAVE1(3),WAVE2(3),LMAX
0043100       COMMON/CGC/C(3,3,100,100)
0043200       COMMON/HAR/Y(100,100)
0043300       COMMON/TEN/TB(3,3,100,100)
0043400       COMMON/COORDS/R,CT,ST,CP,SP,P
0043500       COMPLEX TB,Y,T(3,3,100,100),AIM,EIP,CEIP,STP,CSTP,CTP,CCTP
0043600       DO 2 NL=1,LMAX
0043700       MTOP=NL+1
0043800       DO 3 NM=1,MTOP
0043900       T(1,1,NL,NM)=C(1,1,NL,NM)*Y(NL,NM+2)
0044000       T(1,2,NL,NM)=C(1,2,NL,NM)*Y(NL+1,NM+2)
0044100       T(1,3,NL,NM)=C(1,3,NL,NM)*Y(NL+2,NM+2)
0044200       T(2,1,NL,NM)=C(2,1,NL,NM)*Y(NL,NM+1)
0044300       T(2,2,NL,NM)=C(2,2,NL,NM)*Y(NL+1,NM+1)
0044400       T(2,3,NL,NM)=C(2,3,NL,NM)*Y(NL+2,NM+1)
0044500       T(3,1,NL,NM)=C(3,1,NL,NM)*Y(NL,NM)
0044600       T(3,2,NL,NM)=C(3,2,NL,NM)*Y(NL+1,NM)
0044700       T(3,3,NL,NM)=C(3,3,NL,NM)*Y(NL+2,NM)
0044800     3 CONTINUE
0044900     2 CONTINUE
0045000       R2=0.7071067
0045100       AIM=(0.0,1.0)
0045200       EIP=CMPLX(CP,SP)*R2
0045300       CEIP=CONJG(EIP)
0045400       STP=ST*EIP
0045500       CSTP=CONJG(STP)
0045600       CTP=CT*EIP
0045700       CCTP=CONJG(CTP)
0045800       DO 5 NL=1,LMAX
0045900       MTOP=NL+1
```

```
0046000       DO 6 NM=1,MTOP
0046100       TB(1,1,NL,NM)=-STP*T(3,1,NL,NM)+CSTP*T(1,1,NL,NM)+CT*T(2,1,NL,NM)
0046200       TB(2,1,NL,NM)=-CTP*T(3,1,NL,NM)+CCTP*T(1,1,NL,NM)-ST*T(2,1,NL,NM)
0046300       TB(3,1,NL,NM)=-AIM*(EIP*T(3,1,NL,NM)+CEIP*T(1,1,NL,NM))
0046400       TB(1,2,NL,NM)=-STP*T(3,2,NL,NM)+CSTP*T(1,2,NL,NM)+CT*T(2,2,NL,NM)
0046500       TB(2,2,NL,NM)=-CTP*T(3,2,NL,NM)+CCTP*T(1,2,NL,NM)-ST*T(2,2,NL,NM)
0046600       TB(3,2,NL,NM)=-AIM*(EIP*T(3,2,NL,NM)+CEIP*T(1,2,NL,NM))
0046700       TB(1,3,NL,NM)=-STP*T(3,3,NL,NM)+CSTP*T(1,3,NL,NM)+CT*T(2,3,NL,NM)
0046800       TB(2,3,NL,NM)=-CTP*T(3,3,NL,NM)+CCTP*T(1,3,NL,NM)-ST*T(2,3,NL,NM)
0046900       TB(3,3,NL,NM)=-AIM*(EIP*T(3,3,NL,NM)+CEIP*T(1,3,NL,NM))
0047000     6 CONTINUE
0047100     5 CONTINUE
0047200       RETURN
0047300       END
0047400       SUBROUTINE CLG
0047500       COMMON/PROPS/PI,PMB,RA,REFRAC(3),WAVE1(3),WAVE2(3),LMAX
0047600       COMMON/CGC/C(3,3,100,100)
0047700       DO 1 NL=1,LMAX
0047800       L=NL
0047900       NP=L+1
0048000       FL=FLOAT(L)
0048100       D1=FL+1.
0048200       D2=D1+FL+1.
0048300       D3=D2+1.
0048400       D4=2.*FL-1.
0048500       DO 1 NM=1,NP
0048600       M=NM-1
0048700       FM=FLOAT(M)
0048800       FLM=FL-FM
0048900       FPM=FL+FM
0049000       FPM1=FPM+1.
0049100       FLM1=FLM+1.
0049200       FPM2=FPM-1.
0049300       FLM2=FLM-1.
0049400       IF(L-M)2,2,3
0049500     3 C(1,1,NL,NM)=SQRT(FLM*FLM2/2./FL/D4)
0049600     2 C(2,1,NL,NM)=SQRT(FPM*FLM/FL/D4)
0049700       C(3,1,NL,NM)=SQRT(FPM*FPM2/2./FL/D4)
0049800       C(1,2,NL,NM)=SQRT(FLM*FPM1/2./FL/D1)
0049900       C(2,2,NL,NM)=FM/SQRT(FL*D1)
0050000       C(3,2,NL,NM)=-SQRT(FPM*FLM1/2./FL/D1)
0050100       C(1,3,NL,NM)=SQRT(FPM1*(FPM1+1.)/D3/D2)
0050200       C(2,3,NL,NM)=-SQRT(FLM1*FPM1/D1/D3)
0050300       C(3,3,NL,NM)=SQRT((FLM1+1.)*FLM1/D3/D2)
0050400     1 CONTINUE
0050500       RETURN
0050600       END
0050700       BLOCK DATA
0050800       COMMON/PROPS/PI,PMB,RA,REFRAC(3),WAVE1(3),WAVE2(3),LMAX
0050900       DATA REFRAC/1.5213,1.5170,1.5255/
0051000       DATA WAVE1/0.445,0.465,0.426/
0051100       DATA PI/3.1415926536/
0051200       DATA PMB/1.0/
0051300       END
0051400 /EOF
EOF
```



FIGURE 1



FIGURE 2



FIGURE 3



FIGURE 4



FIGURE 5

ORIGINAL PAGE IS
OF POOR QUALITY



FIGURE 6



FIGURE 7



FIGURE 8. Detected Intensity at 505nm as the Dye Laser Wavelength was Varied.